Université Grenoble Alpes (UGA)

UFR en Informatique, Mathématique et Mathématiques Appliquées (IM2AG)
Département Licence Sciences et Technologies (DLST)

Architectures des ordinateurs (une introduction)

Unité d'Enseignement INF401 pour les Parcours INM, MIN et MIN-int

Année 1, l'exécution des programmes en langage machine.

(comprendre pour programmer efficacement et sans bug)

Documentation Technique

Année Universitaire 2025 / 2026

Table des matières

Première partie

Documentation Technique

Chapitre 1

Environnement informatique pour les travaux pratiques

1.1 Serveurs d'exécution

Un serveur Linux : `turing` est disponible pour les TPs de INF401 (im2ag-turing.univ-grenoble-alpes.fr).

Depuis un poste du DLST : utiliser un logiciel de connexion à distance (typiquement `MobaXterm`, `putty`, `xming`, `wsl` ou `cygwin/X`) disponible sur le pc windows pour obtenir une bannière de connexion sur turing.

Depuis une machine de type POSIX (unix, linux, macOS, winOs/WSL) : lancer la commande `ssh -X -C im2ag-turing.univ-grenoble-alpes.fr` dans un terminal (Xterm, etc) pour vous connecter.

Des exercices d'accompagnement avec serveur d'exécution et autoévaluation sont également disponibles sur la plateforme Caseine (`https://moodle.caseine.org/course/view.php?id=716`)

1.2 Emplacement des fichiers

Les fichiers nécessaires pour effectuer chaque TP sont situés dans le répertoire : `/Public/401_INF_Public/TP`$<i>$ alias de `/home/Public/401_INF_Public/TP`$<i>$, où $<i>$ désigne le numéro du TP.
Par exemple, les fichiers nécessaires pour la première séance sont situés dans le répertoire `/Public/401_INF_Public/TP1`. Lorsque le TP dure plus d'une séance le nom du répertoire porte les numéros des deux séances, comme par exemple `/Public/401_INF_Public/TP3et4`.

1.3 Configuration de la session de travail

Les opérations suivantes doivent être effectuées pour configurer *une fois pour toutes* votre environnement de travail :

1. Se connecter à `turing`
2. Exécuter les commandes de configuration contenues dans le fichier `config.sh` au moyen de la commande : `source /Public/401_INF_Public/config.sh`

   Vérifier que le répertoire `/opt/gnu/arm/bin` est bien en tête du chemin d'accès aux exécutables, au moyen de la commande : `echo $PATH`

Cette opération installe une fois pour toutes l'environnement requis pour les TPs. Elle n'est à exécuter qu'une seule fois. *Elle n'aura pas à être ré-exécutée lors des autres séances.*

Votre binôme doit ensuite répéter la même opération, afin que vous puissiez tous deux travailler avec un environnement correct dans la suite du semestre :

— Il doit se connecter à son tour à `turing` (depuis un autre poste ou sur le même après que vous vous soyez vous-même déconnecté).
— Il doit ensuite exécuter sous son identité la commande :
`source /Public/401_INF_Public/config.sh`

1.4 Démarrage d'une session de travail

Les opérations suivantes sont à effectuer **au début de chaque séance** :

1. Se connecter à `turing`.
2. Ouvrir une deuxième fenêtre au moyen de la commande : `xterm &` (Ctrl+clic central ou droit pour options de configuration).
3. Copier le répertoire `/Public/401_INF_Public/TP`<*i*> dans votre répertoire de travail. Par exemple, pour le 1^{er} TP utilisez la commande :
`cp -r /Public/401_INF_Public/TP1 .`
(*ne pas oublier le point à la fin de la commande précédente* !...)
Puis pour pouvoir modifier vos fichiers : `chmod -R u+w TP1`
4. Effectuer les exercices décrits dans l'énoncé du TP.

1.5 TP : ressources disponibles

1.5.1 Outils

— `nedit` : création et modification de fichiers au format texte (`gedit` également disponible)..
— `cat` et `less` : visualisation de fichiers texte.
— `bitmap` et `bmtoa` : affichage d'une image monochrome au format *bitmap*.
— `xli` : visualisation de fichiers contenant une image.
— `hexdump` : visualisation en hexadécimal d'un fichier binaire ou texte.
— `arm-eabi-gcc` : compilateur C et assembleur pour processeur Arm.
— `arm-eabi-objdump` : utilitaire permettant d'observer le contenu d'un fichier binaire ayant été produit par `arm-eabi-gcc`.

1.6 Quelques commandes utiles (rappels)

— Créer une copie d'un fichier existant (sans utiliser `nedit` :
`cp` <*nom fichier original*> <*nom nouveau fichier*>
Exemple : `cp fich1.c fich2.c`
ou bien :
`cp` <*nom fichier original*> <*nom répertoire destination*>
Exemple : `cp fich1.c repA`

— Renommer un fichier :
`mv` <*nom original du fichier*> <*nouveau nom du fichier*>
Exemple : `mv fich1.c fich2.c`

— Déplacer un fichier :
`mv` <*nom du fichier*> <*nom du répertoire destination*>
Exemple : `mv fich1.c ../repB`

— Afficher (sans le modifier) le contenu d'un *gros* fichier (inutile d'utiliser `nedit`) :
`less` <*nom du fichier*> Exemple : `less fich1.c`

— Afficher (sans le modifier) le contenu d'un *petit* fichier (inutile d'utiliser `nedit`) :
`cat` <*nom du fichier*> Exemple : `cat fich2.s`

1.7 Commandes raccourcies

— `arm-eabi-gcc -c` ou `armas` <*nom du fichier source .s*>
Assemblage d'un fichier source. Le résultat est un fichier binaire translatable dont le nom est suffixé par `.o` (`prog.o` dans l'exemple).
Exemple : `armas prog.s`

— `arm-eabi-gcc -c` ou `armcc` <*nom du fichier source .c*>
Compilation d'un fichier en langage C. Le résultat est un fichier binaire translatable dont le nom est suffixé par `.o` (`prog.o` dans l'exemple).
Exemple : `armcc prog.c`

— `arm-eabi-gcc -o` ou `armbuild` <*nom du fichier exécutable*> <*nom du fichier source*> [<*liste de fichiers `.o` à ajouter éventuellement*>]
Assemblage (ou compilation, pour un fichier en langage C) *et* production d'un exécutable.
Exemple 1 : `armbuild prog1 prog1.s lib.o`
Exemple 2 : `armbuild prog2 prog2.c`

— `arm-eabi-run` ou `armrun` <*nom du fichier exécutable*>
Exécution/simulation d'un fichier binaire exécutable.
Exemple : `armrun prog`

— `arm-eabi-gdb` ou `armgdb` (ou `armddd`) <*nom du fichier exécutable*>
Mise au point d'un fichier binaire exécutable (mode graphique).
Exemple : `armgdb prog`

— `arm-eabi-objdump -j .data -s` ou `armdata` <*nom du fichier "objet" .o*>
Observation de la section .data d'un fichier binaire translatable. Le résultat est affiché à l'écran.
Exemple : `armdata prog.o`

— `arm-eabi-objdump -S` ou `armdisas` <*nom du fichier "objet" .o*>
Observation/désassemblage de la section .text d'un fichier binaire translatable. Le résultat est affiché à l'écran.
Exemple : `armdisas prog.o`

Chapitre 2

Langage machine et langage d'assemblage ARM

2.1 Résumé de documentation technique ARMv4 pour ARM7TDMI

2.1.1 Organisation des registres

Dans le mode dit "utilisateur" le processeur `ARM` a 16 registres visibles de taille 32 bits nommés `r0, r1, ..., r15` :

- `r13` (synonyme `sp`, comme "stack pointer") est utilisé comme registre pointeur de pile.
- `r14` (synonyme `lr` comme "link register") est utilisé par l'instruction "branch and link" (`bl`) pour sauvegarder l'adresse de retour lors d'un appel de procédure.
- `r15` (synonyme `pc`, comme "program counter") est le registre compteur de programme.

Les conventions de programmation des procédures (ATPCS="ARM-Thumb Procedure Call Standard, Cf. Developer Guide, chapitre 2) précisent :

- les registres `r0, r1, r2 et r3` sont utilisés pour le passage des paramètres (données ou résultats ; en cas d'appel de fonction, r0 est la valeur renvoyée ; au delà, les paramètres sont placés sur la pile)
- le registre `r12` (synonyme `ip`) est un "intra-procedure call scratch register" ; autrement dit il peut être modifié par une procédure appelée.
- le compilateur `arm-eabi-gcc` utilise le registre `r11` (synonyme `fp` comme "frame pointer") comme base de l'environnement de définition d'une procédure.

Par déduction,

- les registres `r4, r5, ..., r10` doivent pouvoir être utilisés librement par le programmeur.

La figure 2.1 présente un récapitulatif des différents registres disponibles [1].

En particulier, la figure résume comment ces registres sont utilisés lors d'un appel de procédure, et quels registres doivent être sauvés par la fonction appelante (c'est à dire, que la fonction appelante ne doit pas s'attendre à retrouver inchangés), et quels registres doivent être sauvés par la fonction appelée (c'est à dire, que la fonction appelante s'attend à retrouver en l'état après un appel de fonction).

Le processeur a de plus un registre d'état, `cpsr` pour "Current Program Status Register", qui comporte entre autres les codes de conditions arithmétiques. Le registre d'état est décrit dans la figure 2.2.

Les bits `N, Z, C` et `V` sont les codes de conditions arithmétiques, `I` et `F` permettent le masquage des interruptions et `mode` définit le mode d'exécution du processeur (`User`, `Abort`, `Supervisor`, `IRQ`, etc).

1. Basé sur la documentation officielle : `https://github.com/ARM-software/abi-aa`

Registre	Spécial	Sauvegarde	Rôle dans l'appel de procédure
`r15`	`pc`	-	*Program Counter*
`r14`	`lr`	appelée	*Link Register*
`r13`	`sp`	appelée	*Stack Pointer*
`r12`	`ip`	appelée	*Intra-Procedure-call scratch register*
`r11`	`fp`	appelée	*Frame Pointer* ou variable 8
`r10`		appelée	variable 7
`r9`		appelée	variable 6
`r8`		appelée	variable 5
`r7`		appelée	variable 4
`r6`		appelée	variable 3
`r5`		appelée	variable 2
`r4`		appelée	variable 1
`r3`		appelante	argument 4
`r2`		appelante	argument 3
`r1`		appelante	argument 2 / résultat
`r0`		appelante	argument 1 / résultat

FIGURE 2.1 – Récapitulatif des registres du processeur ARM



FIGURE 2.2 – Registre d'état du processeur ARM

2.1.2 Les instructions

Nous utilisons trois types d'instructions : les instructions arithmétiques et logiques (paragraphe 2.1.5), les instructions de rupture de séquence (paragraphe 2.1.6) et les instructions de transfert d'information entre les registres et la mémoire (paragraphe 2.1.7).

Les instructions sont codées sur 32 bits.

Certaines instructions peuvent modifier les codes de conditions arithmétiques `N, Z, C, V` en ajoutant un `S` au nom de l'instruction.

Toutes les instructions peuvent utiliser les codes de conditions arithmétiques en ajoutant un mnémonique (Cf. figure 2.3) au nom de l'instruction. Au niveau de l'exécution, l'instruction est exécutée si la condition est vraie.

2.1.3 Les codes de conditions arithmétiques

La figure 2.3 décrit l'ensemble des conditions arithmétiques.

Toute instruction peut être exécutée sous une des conditions décrites dans la figure 2.3. Le code de la condition figure dans les bits 28 à 31 du code de l'instruction. Par défaut, la condition est `AL`, elle peut être omise : en l'absence de toute autre condition, c'est elle qui sera utilisée, i.e. elle est implicite.

2.1.4 Description de l'instruction de chargement d'un registre

Nous choisissons dans ce paragraphe de décrire en détail le codage d'une instruction.

L'instruction `MOV` permet de charger un registre avec une valeur immédiate ou de transférer la valeur d'un registre dans un autre avec modification par translation ou rotation de cette valeur.

code	mnémonique	signification	condition testée
0000	EQ	égal	Z
0001	NE	non égal	$\overline{Z}$
0010	CS/HS	$\geq$ dans `N`	C
0011	CC/LO	$<$ dans `N`	$\overline{C}$
0100	MI	moins	N
0101	PL	plus	$\overline{N}$
0110	VS	débordement	V
0111	VC	pas de débordement	$\overline{V}$
1000	HI	$>$ dans `N`	$C \wedge \overline{Z}$
1001	LS	$\leq$ dans `N`	$\overline{C} \vee Z$
1010	GE	$\geq$ dans `Z`	$(N \wedge V) \vee (\overline{N} \wedge \overline{V})$
1011	LT	$<$ dans `Z`	$(N \wedge \overline{V}) \vee (\overline{N} \wedge V)$
1100	GT	$>$ dans `Z`	$\overline{Z} \wedge ((N \wedge V) \vee (\overline{N} \wedge \overline{V}))$
1101	LE	$\leq$ dans `Z`	$Z \vee (N \wedge \overline{V}) \vee (\overline{N} \wedge V)$
1110	AL	toujours	

FIGURE 2.3 – Codes des conditions arithmétiques

La syntaxe de l'instruction de transfert est : `MOV [<COND>] [S] <rd>, <opérande>` où `rd` désigne le registre destination et `opérande` est décrit par la table ci-dessous :

`opérande`	commentaire
`#immédiate-8`	entier sur 32 bits (Cf. remarque ci-dessous)
`rm`	registre
`rm, shift #shift-imm-5`	registre dont la valeur est décalée d'un nombre de positions représenté sur 5 bits
`rm, shift rs`	registre dont la valeur est décalée du nombre de positions contenu dans le registre `rs`

Dans la table précédente le champ `shift` de l'opérande peut être `LSL, LSR, ASR, ROR` qui signifient respectivement "logical shift left", "logical shift right", "arithmetic shift right", "rotate right".

Une valeur immédiate est notée selon les mêmes conventions que dans le langage C ; ainsi elle peut être décrite en décimal (15), en hexadécimal (0xF) ou en octal (O17).

Le codage de l'instruction `MOV` est décrit dans les figures 2.4 et 2.5. `<COND>` désigne un mnémonique de condition ; s'il est omis la condition est `AL`. Le bit `S` est mis à 1 si l'on souhaite une mise à jour des codes de conditions arithmétiques. Le bit `I` vaut 1 dans le cas de chargement d'une valeur immédiate. Les codes des opérations `LSL, LSR, ASR, ROR` sont respectivement : $00, 01, 10, 11$.

Remarque concernant les valeurs immédiates : Une valeur immédiate sur 32 bits (opérande `#immediate`) sera codée dans l'instruction au moyen, d'une part d'une constante exprimée sur 8 bits (bits 7 à 0 de l'instruction, figure 2.5, 1^{er} cas), et d'autre part d'une rotation exprimée sur 4 bits (bits 11 à 8) qui sera appliquée à la dite constante lors de l'exécution de l'instruction.

La valeur de rotation, comprise entre 0 et 15, est multipliée par 2 lors de l'exécution et permet donc d'appliquer à la constante une rotation à **droite** d'un nombre pair de positions compris entre 0 et 30. La rotation s'applique aux 8 bits placés initialement à droite dans un mot de 32 bits (qui n'est pas celui qui contient l'instruction).

Il en résulte que ne peuvent être codées dans l'instruction toutes les valeurs immédiates sur 32 bits, seulement celles de la forme immediate-8 Rotate_Right (rotate-imm*2).

Une rotation nulle permettra de coder toutes les valeurs immédiates sur 8 bits.



FIGURE 2.4 – Codage de l'instruction `mov`



FIGURE 2.5 – Codage de la partie `opérande` d'une instruction

Exemples d'utilisations de l'instruction `mov`

```
MOV r4, #42         @ r4 <-- 42
MOV r4, r5          @ r4 <-- r5
MOV r8, r7, LSL #28 @ r8 <-- r7 décalé à gauche de 28 positions
MOV r4, r5, LSR r6  @ r4 <-- r5 décalé à droite de n pos., r6=n
MOVS r7, #-5        @ r7 <-- -5 + positionnement N, Z, C et V
```

2.1.5 Description des instructions arithmétiques et logiques

Les instructions arithmétiques et logiques ont pour syntaxe :
`code-op[<cond>][s] <rd>, <rn>, <opérande>`, où `code-op` est le nom de l'opération, `rn` et `opérande` sont les deux opérandes et `rd` le registre destination.

Le codage d'une telle instruction est donné dans la figure 2.6. `opérande` est décrit dans le paragraphe 2.1.4, figure 2.5.



FIGURE 2.6 – Codage d'une instruction arithmétique ou logique

La table ci-dessous donne la liste des intructions arithmétiques et logiques ainsi que les instructions de chargement d'un registre. Les instructions `TST, TEQ, CMP, CMN` n'ont pas de registre destination, elles ont ainsi seulement deux opérandes ; elles provoquent systématiquement la mise à jour des codes de conditions arithmétiques (dans le codage de l'instruction les bits 12 à 15 sont mis à zéro). Les instructions `MOV` et `MVN` ont un registre destination et un opérande (dans le codage de l'instruction les bits 16 à 19 sont mis à zéro).

code-op	Nom	Explication du nom	Opération	remarque
0000	`AND`	AND	et bit à bit	
0001	`EOR`	Exclusive OR	ou exclusif bit à bit	
0010	`SUB`	SUBstract	soustraction	
0011	`RSB`	Reverse SuBstract	soustraction inversée	
0100	`ADD`	ADDition	addition	
0101	`ADC`	ADdition with Carry	addition avec retenue	
0110	`SBC`	SuBstract with Carry	soustraction avec emprunt	
0111	`RSC`	Reverse Substract with Carry	soustraction inversée avec emprunt	
1000	`TST`	TeST	et bit à bit	pas `rd`
1001	`TEQ`	Test EQuivalence	ou exclusif bit à bit	pas `rd`
1010	`CMP`	CoMPare	soustraction	pas `rd`
1011	`CMN`	CoMpare Not	addition	pas `rd`
1100	`ORR`	OR	ou bit à bit	
1101	`MOV`	MOVe	copie	pas `rn`
1110	`BIC`	BIt Clear	et not bit à bit	
1111	`MVN`	MoVe Not	not (complément à 1)	pas `rn`

Exemples d'utilisations

```
ADD r6, r8, r5           @ r6 <-- r8 + r5
SUB r5, r7, r9           @ r5 <-- r7 - r9
CMP r4, r5               @ calcul de r4-r5 et positionnement NZCV
TST r4, #1               @ calcul de r4 ET 1 et positionnement NZCV
ANDS r8, r7, #0x0000ff00 @ r8 <-- r7 ET 0x0000ff00 et positionnement NZCV
```

2.1.6 Description des instructions de rupture de séquence

Nous utilisons quatre instructions de rupture de séquence : `B[<cond>] <déplacement>`, `BX[<cond>] <registre>`, `BL[<cond>] <déplacement>`, `BLX[<cond>] <registre>`.

a) Instruction `B[<cond>] <déplacement>` L'instruction `BCond` provoque la modification du compteur de programme si la condition est vraie ; le texte suivant est extrait de la documentation ARM :

```
if ConditionPassed(cond) then
   PC <-- PC + (SignExtend(déplacement) << 2)
```

L'expression `(SignExtend(déplacement) << 2)` signifie que le `déplacement` est tout d'abord étendu de façon signée à 32 bits puis mutiplié par 4. Le `déplacement` est en fait un entier relatif (codé sur 24 bits comme indiqué ci-dessous) et qui représente le nombre d'instructions (en avant ou en arrière) entre l'instruction de rupture de séquence et la cible de cette instruction.

Dans le calcul du déplacement, il faut prendre en compte le fait que lors de l'exécution d'une instruction, le compteur de programme ne repère pas l'instruction courante mais deux instructions en avant.



FIGURE 2.7 – Codage de l'instruction de rupture de séquence `b{cond}`

Exemples d'utilisations

```
BEQ +5      @ si cond(EQ) alors pc <-- pc + 4*5
B -8        @ pc <-- pc - 4*8
B Suite     @ PC <-- pc + déplacement (pour aller à l'étiquette Suite)
```

Dans la pratique, on utilise une étiquette (Cf. paragraphe 2.2.4) pour désigner l'instruction cible d'un branchement. C'est le traducteur (i.e. l'assembleur) qui effectue le calcul du déplacement.

La figure 2.8 résume l'utilisation des instructions de branchements conditionnels après une comparaison.

Conditions des instructions de branchement conditionnel				
Type	Entiers relatifs ($\mathbb{Z}$)		Naturels ($\mathbb{N}$) et adresses	
Instruction C	Bxx	Condition	Bxx	Condition
goto	B ou BAL	1110	B ou BAL	1110
if (x== y) goto	BEQ	0000	BEQ	0000
if (x != y) goto	BNE	0001	BNE	0001
if (x < y) goto	BLT	1011	BLO, BCC	0011
if (x<= y) goto	BLE	1101	BLS	1001
if (x > y) goto	BGT	1100	BHI	1000
if (x>= y) goto	BGE	1010	BHS,BCS	0010

FIGURE 2.8 – Utilisation des branchements conditionnels après une comparaison

b) Instruction `BX[<cond>] <registre>` L'instruction `BX Rm` provoque la modification du compteur de programme ; le texte suivant est extrait de la documentation ARM :

```
PC <-- Rm
```

Attention : Dans le cadre des TP, l'adresse passée en paramètre à `BX` doit être paire (cf. infra).

Exemple d'utilisation

```
BX LR       @ pc <-- lr
```

c) Instruction `BL[<cond>] <déplacement>` L'instruction `BL` provoque la modification du compteur de programme avec sauvegarde de l'adresse de l'instruction suivante (appelée `adresse de retour`) dans le registre `lr` ; le texte suivant est extrait de la documentation ARM :

```
lr <-- address of the instruction after the branch instruction
PC <-- PC + (SignExtend(déplacement) << 2)
```



FIGURE 2.9 – Codage de l'instruction de branchement à un sous-programme `bl`

Exemples d'utilisations

```
BL 42     @ lr <-- pc+4 ; pc <-- pc +4*42
BL Fact   @ lr <-- pc+4 ; pc <-- pc + déplacement (pour aller à l'étiquette Fact)
```

d) Instruction `BLX[<cond>] <registre>` L'instruction `BLX Rm` provoque la modification du compteur de programme avec sauvegarde de l'adresse de l'instruction suivante (appelée `adresse de retour`) dans le registre `lr` ; le texte suivant est extrait de la documentation ARM :

```
lr <-- address of the instruction after the branch instruction
PC <-- Rm
```

Attention : Dans le cadre des TP, l'adresse passée en paramètre à `BLX` doit être paire. En effet, l'exécution de `BLX` avec une adresse impaire active un mode spécial du processeur (THUMB) avec un autre jeu d'instructions (codées sur 16 bits). Ce type d'erreur peut avoir des effets assez variés en fonction du programme concerné : on peut obtenir un message d'erreur relatif au mode THUMB ou un comportement arbitraire du simulateur.

Exemples d'utilisations

```
BLX R5       @ lr <-- pc+4 ; pc <-- R5
```

Pour désigner une procédure on utilisera une étiquette ; des exemples sont donnés dans le paragraphe 2.2.4.

2.1.7 Description des instructions de transfert d'information entre les registres et la mémoire

Transfert entre un registre et la mémoire

L'instruction `LDR` dont la syntaxe est : `LDR <rd>, <mode-adressage>` permet le transfert du mot mémoire dont l'adresse est spécifiée par `mode-adressage` vers le registre `rd`. Nous ne donnons pas le codage de l'instruction `LDR` parce qu'il comporte un grand nombre de cas ; nous regardons ici uniquement les utilisations les plus fréquentes de cette instruction.

Le champ `mode-adressage` comporte, entre crochets, un registre et éventuellement une valeur immédiate ou un autre registre, ceux-ci pouvant être précédés du signe + ou −. Le tableau ci-dessous indique pour chaque cas le mot mémoire qui est chargé dans le registre destination. L'instruction `ldr` permet beaucoup d'autres types de calcul d'adresse qui ne sont pas décrits ici.

mode-adressage	opération effectuée
`[rn]`	rd <– mem [rn]
`[rn, #offset12]`	rd <– mem [rn + offset12]
`[rn, #-offset12]`	rd <– mem [rn - offset12]
`[rn, rm]`	rd <– mem [rn + rm]
`[rn, -rm]`	rd <– mem [rn - rm]

Il existe des variantes de l'instruction `LDR` permettant d'accéder à un octet : `LDRB` ou à un mot de 16 bits : `LDRH`. Et si l'on veut accéder à un octet signé : `LDRSB` ou à un mot de 16 bits signé : `LDRSH`. Ces variantes imposent cependant des limitations d'adressage par rapport aux versions 32 bits (exemple : valeur immédiate codée sur 5 bits au lieu de 12).

Pour réaliser le transfert inverse, registre vers mémoire, on trouve l'instruction `STR` et ses variantes `STRB` et `STRH`. La syntaxe est la même que celle de l'instruction `LDR`. Par exemple, l'instruction `STR rd, [rn]` provoque l'exécution : `MEM [rn] <-- rd`.

Codage des instructions `ldr` et `str` : La figure 2.10 donne un sous-ensemble des règles de codage des instructions `ldr` et `str`, suffisant pour traiter les exercices précédents. On peut par exemple coder : `ldr rd, [rn, +/-déplacement]` ; le bit `U` code le signe du déplacement (1 pour +, 0 pour −) et le bit `L` vaut 1 pour `ldr` et 0 pour `str`.



FIGURE 2.10 – Codage des instructions `ldr` et `str`

Exemples d'utilisations

```
LDR r4, [r5]           @ r4 <-32bits-- Mem [r5]
LDR r4, [r6, #4]       @ r4 <-32bits-- Mem [r6 + 4]
LDR r7, [r5, #-8]      @ r7 <-32bits-- Mem [r5 - 8]
LDR r5, [pc, #48]      @ r5 <-32bits-- Mem [pc + 48]
LDRB r5, [r9]          @ 8bits_poids_faibles (r5) <-- Mem [r9],
                       @ extension aux 32 bits avec des 0
STRH r5, [r4, r6]      @ Mem [r4 + r6] <-16bits-- 16bits_poids_faibles (r5)
```

L'instruction `LDR` est utilisée entre autres pour accéder à un mot de la zone text en réalisant un adressage relatif au compteur de programme. Ainsi, l'instruction `LDR r7, [pc, #depl]` permet de charger dans le registre `r7` avec le mot mémoire situé à une distance `depl` du compteur de programme, c'est-à-dire de l'instruction en cours d'exécution. Ce mode d'adressage nous permet de recupérer l'adresse d'un mot de données (Cf. paragraphe 2.2.4).

Pré décrémentation et post incrémentation

Les instructions `LDR` et `STR` offrent des adressages post-incrémentés et pré-décrémentés qui permettent d'accéder à un mot de la mémoire et de mettre à jour une adresse, en une seule instruction. Cela revient à combiner un accès mémoire et l'incrémentation du pointeur sur celle-ci en une seule instruction.

instruction ARM	équivalent ARM	équivalent C
`LDR r4, [r5, #-4]!`	`SUB r5, r5, #4`	`r4 = *--r5`
	`LDR r4, [r5]`	
`LDR r6, [r7], #4`	`LDR r6, [r7]`	`r6 = *r7++`
	`ADD r7, r7, #4`	
`STR r5, [r6, #-4]!`	`SUB r6, r6, #4`	
	`STR r5, [r6]`	
`STR r5, [r7], #4`	`STR r5, [r7]`	
	`ADD r7, r7, #4`	

La valeur à incrémenter ou décrémenter (4 dans les exemples ci-dessus) peut aussi être donnée dans un registre.

Transfert multiples

Le processeur ARM possède des instructions de transfert entre un ensemble de registres et un bloc de mémoire repéré par un registre appelé registre de base : `LDM` et `STM`. Par exemple, `STMFD r7!, {r0,r1,r5}` range le contenu des registres `r0, r1` et `r5` dans la mémoire et met à jour

le registre r7 après le transfert (i.e. r7 = r7 - 12) ; après l'exécution de l'instruction MEM[r7 à jour] contient r0 et MEM[r7 à jour + 8] contient r5.

Il existe 8 variantes de chacune des instructions LDM et STM selon que :

— les adresses de la zone mémoire dans laquelle sont copiés les registres croissent (Increment) ou décroissent (Decrement).

— l'adresse contenue dans le registre de base est incrémentée ou décrémentée avant (Before) ou après (After) le transfert de chaque registre. Notons que l'adresse est décrémentée avant le transfert quand le registre de base repère le mot qui a l'adresse immédiatement supérieure à celle où l'on veut ranger une valeur (Full) ; l'adresse est incrémentée après le transfert quand le registre de base repère le mot où l'on veut ranger une valeur (Empty).

— le registre de base est modifié à la fin de l'exécution quand il est suivi d'un ! ou laissé inchangé sinon.

Ces instructions servent aussi à gérer une pile. Il existe différentes façons d'implémenter une pile selon que :

— le pointeur de pile repère le dernier mot empilé (Full) ou la première place vide (Empty).

— le pointeur de pile progresse vers les adresses basses quand on empile une information (Descending) ou vers les adresses hautes (Ascending).

Par exemple, dans le cas où le pointeur de pile repère l'information en sommet de pile (case pleine) et que la pile évolue vers les adresses basses (lorsque l'on empile l'addresse décroit), on parle de pile Full Descending et on utilise l'instruction STMFD pour empiler et LDMFD pour dépiler.

Les modes de gestion de la pile peuvent être caractérisés par la façon de modifier le pointeur de pile lors de l'empilement d'une valeur ou de la récupération de la valeur au sommet de la pile. Par exemple, dans le cas où le pointeur de pile repère l'information en sommet de pile et que la pile évolue vers les adresses basses, pour empiler une valeur il faut décrémenter le pointeur de pile avant le stockage en mémoire ; on utilisera l'instruction STMDB (Decrement Before). Dans le même type d'organisation pour dépiler on accède à l'information au sommet de pile puis on incrémente le pointeur de pile : on utilise alors l'instruction LDMIA (Increment After).

Selon que l'on prend le point de vue gestion d'un bloc de mémoire repéré par un registre ou gestion d'une pile repérée par le registre pointeur de pile, on considère une instruction ou une autre ... Ainsi, les instructions STMFD et STMDB sont équivalentes ; de même pour les instructions LDMFD et LDMIA.

Les tables suivantes donnent les noms des différentes variantes des instructions LDM et STM, chaque variante ayant deux noms synonymes l'un de l'autre.

nom de l'instruction	synonyme
LDMDA (decrement after)	LDMFA (full ascending)
LDMIA (increment after)	LDMFD (full descending)
LDMDB (decrement before)	LDMEA (empty ascending)
LDMIB (increment before)	LDMED (empty descending)

nom de l'instruction	synonyme
STMDA (decrement after)	STMED (empty descending)
STMIA (increment after)	STMEA (empty ascending)
STMDB (decrement before)	STMFD (full descending)
STMIB (increment before)	STMFA (full ascending)

La figure 2.11 donne un exemple d'utilisation.

Exemples d'utilisations pour une pile FULL Descending

```
STMFD r7!, {r4,r5,r6}    @ cf. schéma
```



FIGURE 2.11 – Transfert multiples mémoire/registres : `STMFD r7!, {r4,r5,r6}` ou (`STMDB ...`) permet de passer de l'état `a)` de la mémoire à l'état `b)`. `LDMFD r7!, {r4,r5,r6}` (ou `LDMIA ...`) réalise l'inverse.

```
PUSH {lr}                @ équivalent de STMFD SP!, {LR}, empile lr
LDMFD r7!, {r4,r5,r6}    @ cf. schéma
POP {R6}                 @ équivalent de LDMFD SP!, {r7}, dépile r6
```

2.2 Langage d'assemblage

2.2.1 Structure d'un programme en langage d'assemblage

Un programme est composé de trois types de sections :
— données intialisées ou non (`.data`)
— données non initialisées (`.bss`)
— instructions (`.text`)

Les sections de données sont optionnelles, celle des instructions est obligatoire. On peut écrire des commentaires entre le symbole `@` et la fin de la ligne courante. Ainsi un programme standard a la structure :

```
   .data
@ déclaration de données
@ ...

   .text
@ des instructions
@ ...
```

2.2.2 Déclaration de données

Le langage permet de déclarer des valeurs entières en décimal (éventuellement précédées de leur signe) ou en hexadécimal ; on précise la taille souhaitée.

Exemple :

```
    .data
  .word 4536   @ déclaration de la valeur 4536 sur 32 bits (1 mot)
  .hword -24   @ déclaration de la valeur -24 sur 16 bits (1 demi mot)
  .byte 5      @ déclaration de la valeur 5 sur 8 bits (1 octet)
  .word 0xfff2a35f @ déclaration d'une valeur en hexadécimal sur 32 bits
  .byte 0xa5   @ idem sur 8 bits
```

On peut aussi déclarer des chaînes de caractères suivies ou non du caractère de code ASCII 00. Un caractère est codé par son code ASCII (Cf. paragraphe II).

Exemple :

```
    .data
  .ascii "un texte" @ déclaration de 8 caractères...
  .asciz "un texte" @ déclaration de 9 caractères, les mêmes que ci-dessus
                    @ plus le code 0 à la fin
```

La définition de données doit respecter les règles suivantes, qui proviennent de l'organisation physique de la mémoire :

— un mot de 32 bits doit être rangé à une adresse multiple de 4
— un mot de 16 bits doit être rangé à une adresse multiple de 2
— il n'y a pas de contrainte pour ranger un octet (mot de 8 bits)

Pour recadrer une adresse (en section data ou bss) le langage d'assemblage met à notre disposition la directive `.balign`. Lorsqu'une directive de réservation de g octets suit une réservation de taille inférieure une directive .balign g sera insérée entre les deux. La section text peut aussi être concernée (cas d'une constante chaîne de caractères suivie d'une instruction).

Exemple :

```
   .data
@ on note AD l'adresse de chargement de la zone data
@ que l'on suppose multiple de 4 (c'est le cas avec les outils utilisés)
  .hword 43    @ après cette déclaration la prochaine adresse est AD+2
  .balign 4    @ recadrage sur une adresse multiple de 4
  .word 0xffff1234  @ rangé à l'adresse AD+4
  .byte 3      @ après cette déclaration la prochaine adresse est AD+9
  .balign 2    @ recadrage sur une adresse multiple de 2
  .hword 42    @ rangé à l'adresse AD+10
```

On peut aussi réserver de la place en zone `.data` ou en zone `.bss` avec la directive `.skip`.
`.skip 256` réserve 256 octets qui ne sont pas initialisés lors de la réservation. On pourra par programme écrire dans cette zone de mémoire.

2.2.3 La zone text

Le programmeur y écrit des instructions qui seront codées par l'assembleur (le traducteur) selon les conventions décrites dans le paragraphe 2.1.

La liaison avec le système (chargement et lancement du programme) est réalisée par la définition d'une étiquette (Cf. paragraphe suivant) réservée : `main`.

Ainsi la zone `text` est :

```
  .text
  .global main
main:

@ des instructions ARM
@ ...
```

2.2.4 Utilisation d'étiquettes

Une donnée déclarée en zone `data` ou `bss` ou une instruction de la zone `text` peut être précédée d'une étiquette. Une étiquette représente une adresse et permet de désigner la donnée ou l'instruction concernée.

Les étiquettes représentent une facilité d'écriture des programmes en langage d'assemblage.

Expression d'une rupture de séquence

On utilise une étiquette pour désigner l'instruction cible d'un branchement. C'est le traducteur (i.e. l'assembleur) qui effectue le calcul du déplacement. Par exemple :

```
etiq: MOV r4, #22
      ADD r5, r6, r4
      CMP r5, #13
      BNE etiq
```

Pour une conditionnelle "si r4=0 alors r4=1 sinon r4=2 fin si" deux traductions sont possibles :

```
si:    CMP r4, #0
       BEQ alors
sinon: MOV r4, #2
       B finsi
alors: MOV r4, #1
finsi:
```

```
si:   CMP  r4,#0    @ version préférable
      BNE  sinon    @ respectant l'ordre du code d'origine
alors: MOV  r4,#1
       B    finsi
sinon: MOV r4,#2
finsi:
```

Pour une boucle "répéter r5=r5+1 ; r4=r4-r5 tant que $r4 > 0$ fin répéter" une traduction possible sera donc :

```
corpsboucle: ADD r5, r5, #1
             SUB r4, r4, r5
test:        CMP r4, #0
             BGT corpsboucle
finboucle:
```

Accès à une donnée depuis la zone `text`

```
   .data

X: .word 5

   .text

@ acces au mot d'adresse X
   LDR r5, LD_X  @ r5 <-- l'adresse X
   LDR r6, [r5]    @ r5 <-- Mem[X] c'est-à-dire 5
```

```
   MOV r7, #245    @ r7 <-- 245
   STR r7, [r5]    @ Mem[X] <-- r7
                   @ la mémoire d'adresse X a été modifiée

@ plus loin
LD_X: .word X  @ déclaration de l'adresse X en zone text
```

L'instruction `LDR r5, LD_X` est codée avec un adressage relatif au compteur de programme : `LDR r5, [pc, #depl]` (Cf. paragraphe 2.1.7).

Appel d'une procédure

On utilise l'instruction `BL` lorsque la procédure appelée est désignée directement par une étiquette. Version simple :

```
   ...
ma_proc: @ corps de la procedure
         bx lr
    ...
main:
...
    @ appel de la procedure ma_proc
    BL ma_proc
...
```

Pour une version plus complète :
appelée `ma_proc` :

1. empiler l'adresse de retour (`lr`)
2. empiler la valeur `fp` de l'appelant
3. placer `fp` pour repérer les variables de l'appelée (en général : `MOV FP, SP`)
4. allouer la place pour les variables locales
5. empiler les registres temporaires utilisés
6. **corps de la procédure ou fonction**
7. si fonction, le résultat est rangé en **fp+8**
8. dépiler les registres temporaires utilisés
9. libérer la place allouée aux variables locales
10. dépiler `fp`
11. dépiler l'adresse de retour (`lr`)
12. retour à l'appelant : `bx lr`

appelant `main` :

1. préparer et empiler les paramètres (valeurs et/ou adresses)
2. si fonction, réserver une place dans la pile pour le résultat
3. appeler ma_proc : `BL ma_proc`
4. si fonction, récupérer le résultat
5. libérer la place allouée aux paramètres

6. si fonction, libérer la place allouée au résultat

Organisation de la pile en cours d'exécution du corps de la fonction appelée :



Remarque : On peut utiliser l'instruction `BLX` lorsque l'adresse de la procédure est rangée dans un registre, par exemple lorsqu'une procédure est passée en paramètre :

```
    LDR   r5, L_proc    @ r5 <-- adresse ma_proc
    BLX   r5
    ...
L_proc: .word ma_proc
```

2.3 Organisation de la mémoire : petits bouts, gros bouts

La mémoire du processeur ARM peut être vue comme un tableau d'octets repérés par des numéros appelés **adresse** qui sont des entiers naturels sur 32 bits. On peut ranger dans la mémoire des mots de 32 bits, de 16 bits ou des octets (mots de 8 bits). Le paragraphe 2.2.2 indique comment déclarer de tels mots.

Dans la mémoire les mots de 32 bits sont rangés à des adresses multiples de 4. Il y a deux conventions de rangement de mots en mémoire selon l'ordre des octets de ce mot.

Considérons par exemple le mot `0x12345678`.

— convention dite "Big endian" (Gros bouts) :
les 4 octets `12, 34, 56, 78` du mot `0x12345678` sont rangés aux adresses respectives 4x, 4x+1, 4x+2, 4x+3.

— convention dite "Little endian" (Petits Bouts) :
les 4 octets `12, 34, 56, 78` du mot `0x12345678` sont rangés aux adresses respectives 4x+3, 4x+2, 4x+1, 4x.

Le processeur ARM suit la convention "Little endian". La conséquence est que lorsqu'on lit le mot de 32 bits rangé à l'adresse `4x` on voit : `78563412`, c'est-à-dire qu'il faut lire "à l'envers". Selon les outils utilisés le mot de 32 bits est présenté sous cette forme ou sous sa forme externe, plus agréable...

En général les outils de traduction et de simulation permettent de travailler avec une des deux conventions moyennant l'utilisation d'options particulières lors de l'appel des outils (option `-mbig-endian`).

2.4 Commandes de traduction, exécution, observation

2.4.1 Traduction d'un programme

Pour traduire un programme écrit en C contenu dans un fichier `prog.c` :
`arm-eabi-gcc -g -o prog prog.c`. L'option `-o` permet de préciser le nom du programme exécutable ; `o` signifie "output". L'option `-g` permet d'avoir les informations nécessaires à la mise au point sous débogueur (Cf. paragraphe 2.4.2).

Pour traduire un programme écrit en langage d'assemblage ARM contenu dans un fichier `prog.s` :
`arm-eabi-gcc -Wa,--gdwarf2 -o prog prog.s`.

Lorsque l'on veut traduire un programme qui est contenu dans plusieurs fichiers que l'on devra rassembler (on dit "lier"), il faut d'abord produire des versions partielles qui ont pour suffixe `.o`, le `o` voulant dire ici "objet". Par exemple, on a deux fichiers : principal.s et biblio.s, le premier contenant l'étiquette `main`. On effectuera la suite de commandes :

```
arm-eabi-gcc -c -Wa,--gdwarf2 biblio.s
arm-eabi-gcc -c -Wa,--gdwarf2 principal.s
arm-eabi-gcc -g -o prog principal.o biblio.o
```

La première produit le fichier `biblio.o`, la seconde produit le fichier `principal.o`, la troisième les relie et produit le fichier exécutable `prog`.

Noter que les deux commandes suivantes ont le même effet :
`arm-eabi-gcc -c prog.s` et
`arm-eabi-as -o prog.o prog.s`. Elles produisent toutes deux un fichier objet `prog.o` sans les informations nécessaires à l'exécution sous débogueur.

2.4.2 Exécution d'un programme

Exécution directe

On peut exécuter un programme directement avec :
`arm-eabi-run prog`. S'il n'y a pas d'entrées-sorties, on ne voit évidemment rien...

Exécution avec un débogueur

Nous pouvons utiliser deux versions du même débogueur : `gdb` et `ddd`. On parle aussi de metteur au point. C'est un outil qui permet d'exécuter un programme instruction par instruction en regardant les "tripes" du processeur au cours de l'exécution : valeur contenues dans les registres, dans le mot d'état, contenu de la mémoire, etc.

`gdb` est la version de base (textuelle), `ddd` est la même mais graphique (avec des fenêtres, des icônes, etc.), elle est plus conviviale mais plus sujette à des problèmes techniques liés à l'installation du logiciel...

Soit le programme objet exécutable : `prog`. Pour lancer `gdb` :
`arm-eabi-gdb prog`. Puis taper successivement les commandes : `target sim` et enfin `load`. Maintenant on peut commencer la simulation.

Pour éviter de taper à chaque fois les deux commandes précédentes, vous pouvez créer un fichier de nom `.gdbinit` dont le contenu est :

```
# un diese débute un commentaire
# commandes de demarrage pour arm-eabi-gdb
target sim
load
```

Au lancement de `arm-eabi-gdb prog`, le contenu de ce fichier sera automatiquement exécuté.

Voilà un ensemble de commandes utiles :

- placer un point d'arrêt sur une instruction précédée d'une étiquette, par exemple : `break main`. On peut aussi demander `break no` avec `no` un numéro de ligne dans le code source. Un raccourci pour la commande est `b`.
- enlever un point d'arrêt : `delete break numéro_du_point_d'arrêt`
- voir le code source : `list`
- lancer l'exécution : `run`
- poursuivre l'exécution après un arrêt : `cont`, raccourci : `c`
- exécuter l'instruction à la ligne suivante, en entrant dans les procédures : `step`, raccourci `s`
- exécuter l'instruction suivante (sans entrer dans les procédures) : `next`, raccourci `n`
- voir la valeur contenue dans les registres : `info reg`
- voir la valeur contenue dans le registre `r1` : `info reg $r1`
- voir le contenu de la mémoire à l'adresse `etiquette` : `x &etiquette`
- voir le contenu de la mémoire à l'adresse `0x3ff5008` : `x 0x3ff5008`
- voir le contenu de la mémoire en précisant le nombre de mots et leur taille.
  `x /nw adr` permet d'afficher `n` mots de 32 bits à partir de l'adresse `adr`.
  `x /ph adr` permet d'afficher `p` mots de 16 bits à partir de l'adresse `adr`.
- modifier le contenu du registre `r3` avec la valeur `0x44` exprimée en hexadécimal : `set $r3=0x44`
- modifier le contenu de la mémoire d'adresse `etiquette` : `set *etiquette = 0x44`
- sortir : `quit`
- La touche `Enter` répète la dernière commande.

Et ne pas oublier : `man gdb` sous Unix (ou Linux) et quand on est sous `gdb` : `help nom_de_commande...`

Pour lancer `ddd` : `ddd --debugger arm-eabi-gdb`. On obtient une grande fenêtre avec une partie dite "source" (en haut) et une partie dite "console" (en bas). Dans la fenêtre "console" taper successivement les commandes : `file prog`, `target sim` et enfin `load`.

On voit apparaître le source du programme en langage d'assemblage dans la fenêtre "source" et une petite fenêtre de "commandes". Maintenant on peut commencer la simulation.

Toutes les commandes de `gdb` sont utilisables soit en les tapant dans la fenêtre "console", soit en les sélectionnant dans le menu adéquat. On donne ci-dessous la description de quelques menus. Pour le reste, il suffit d'essayer.

- placer un point d'arrêt : sélectionner la ligne en question avec la souris et cliquer sur l'icône break (dans le panneau supérieur).
- démarrer l'exécution : cliquer sur le bouton Run de la fenêtre "commandes". Vous voyez apparaître une flèche verte qui vous indique la position du compteur de programme i.e. où en est le processeur de l'exécution de votre programme.
- le bouton Step permet l'exécution d'une ligne de code, le bouton Next aussi mais en entrant dans les procédures et le bouton Cont permet de poursuivre l'exécution.
- enlever un point d'arrêt : se positionner sur la ligne désirée et cliquer à nouveau sur l'icône break.
- voir le contenu des registres : sélectionner dans le menu Status : Registers ; une fenêtre apparaît. La valeur contenue dans chaque registre est donnée en hexadécimal (0x...) et en décimal.

— observer le contenu de la memoire étiquettée `etiquette` : apres avoir sélectionné memory dans le menu Data, on peut soit donner l'adresse en hexadecimal 0x... si on la connaît, soit donner directement le nom `etiquette` dans la case `from` en le précédant du caractère &, c'est-à-dire `&etiquette`.

2.4.3 Observation du code produit

Considérons un programme objet : `prog.o` obtenu par traduction d'un programme écrit en langage C ou en langage d'assemblage. L'objet de ce paragraphe est de décrire l'utilisation d'un ensemble d'outils permettant d'observer le contenu du fichier `prog.o`. Ce fichier contient les informations du programme source codées et organisées selon un format appelé `format ELF`.

On utilise trois outils : `hexdump`, `arm-eabi-readelf`, `arm-eabi-objdump`.

`hexdump` donne le contenu du fichier dans une forme brute.
`hexdump prog.o` donne ce contenu en hexadécimal complété par le caractère correspondant quand une valeur correspond à un code ascii ; de plus l'outil indique les adresses des informations contenues dans le fichier en hexadécimal aussi.

`arm-eabi-objdump` permet d'avoir le contenu des zones `data` et `text` avec les commandes respectives :
`arm-eabi-objdump -j .data -s prog.o` et
`arm-eabi-objdump -j .text -s prog.o`. Ce contenu est donné en hexadécimal. On peut obtenir la zone `text` avec le désassemblage de chaque instruction :
`arm-eabi-objdump -j .text -d prog.o`.

`arm-eabi-readelf` permet d'avoir le contenu du reste du fichier.
`arm-eabi-readelf -a prog.o` donne l'ensemble des sections contenues dans le fichier sauf les zones `data` et `text`.
`arm-eabi-readelf -s prog.o` donne le contenu de la table des symboles.

Deuxième partie

Annexes

Annexe A : Codage ASCII des caractères

Dec	Hex	Char	Dec	Hex	Char	Dec	Hex	Char	Dec	Hex	Char
0	00	NUL	32	20	SPACE	64	40	@	96	60	‘
1	01	SOH	33	21	!	65	41	A	97	61	a
2	02	STX	34	22	”	66	42	B	98	62	b
3	03	ETX	35	23	#	67	43	C	99	63	c
4	04	EOT	36	24	$	68	44	D	100	64	d
5	05	ENQ	37	25	%	69	45	E	101	65	e
6	06	ACK	38	26	&	70	46	F	102	66	f
7	07	BEL	39	27	’	71	47	G	103	67	g
8	08	BS	40	28	(	72	48	H	104	68	h
9	09	HT	41	29	)	73	49	I	105	69	i
10	0A	LF	42	2A	*	74	4A	J	106	6A	j
11	0B	VT	43	2B	+	75	4B	K	107	6B	k
12	0C	FF	44	2C	,	76	4C	L	108	6C	l
13	0D	CR	45	2D	-	77	4D	M	109	6D	m
14	0E	SO	46	2E	.	78	4E	N	110	6E	n
15	0F	SI	47	2F	/	79	4F	O	111	6F	o
16	10	DLE	48	30	0	80	50	P	112	70	p
17	11	DC1	49	31	1	81	51	Q	113	71	q
18	12	DC2	50	32	2	82	52	R	114	72	r
19	13	DC3	51	33	3	83	53	S	115	73	s
20	14	DC4	52	34	4	84	54	T	116	74	t
21	15	NAK	53	35	5	85	55	U	117	75	u
22	16	SYN	54	36	6	86	56	V	118	76	v
23	17	ETB	55	37	7	87	57	W	119	77	w
24	18	CAN	56	38	8	88	58	X	120	78	x
25	19	EM	57	39	9	89	59	Y	121	79	y
26	1A	SUB	58	3A	:	90	5A	Z	122	7A	z
27	1B	ESC	59	3B	;	91	5B	[	123	7B	{
28	1C	FS	60	3C	<	92	5C	\	124	7C	|
29	1D	GS	61	3D	=	93	5D	]	125	7D	}
30	1E	RS	62	3E	>	94	5E	^	126	7E	~
31	1F	US	63	3F	?	95	5F	_	127	7F	DEL

Annexe B : Entiers en base 2, types C

La figure **??** illustre les représentations d'entiers naturels et signés pour une taille de mot de 4 bits. A chaque entier peut être associé un angle. Effectuer une addition revient à ajouter les angles correspondant. Un débordement de produit au-delà d'un demi-tour en arithmétique signée ou d'un tour complet en arithmétique naturelle.

n				2^n		
décimal	hexa	octal	binaire	décimal	hexa	commentaire
0	0	00	0000	1	1	
1	1	01	0001	2	2	
2	2	02	0010	4	4	
3	3	03	0011	8	8	
4	4	04	0100	16	10	un quartet = un chiffre hexa
5	5	05	0101	32	20	
6	6	06	0110	64	40	
7	7	07	0111	128	80	
8	8	10	1000	256	100	un octet = deux chiffres hexa
9	9	11	1001	512	200	
10	A	12	1010	1024	400	$1K_b$
11	B	13	1011	2048	800	$2K_b$
12	C	14	1100	4096	1000	$4K_b$
13	D	15	1101	8192	2000	$8K_b$
14	E	16	1110	16384	4000	$16K_b$
15	F	17	1111	32768	8000	$32K_b$
16	10	20	10000	65536	10000	$64K_b$
20	14	24	10100	1048576	100000	$1M_b = 1K_b^2 = 5$ chiffres
30	1E	36	11110	$\tilde{}1.07 \times 10^9$	40000000	$1G_b = 1K_b^3$

Les tableaux ci-dessus et ci-dessous récapitulent les principales puissances de 2 utiles, avec leur représentation en hexadécimal et les puissances de 10 approchées correspondantes, ainsi que les types C entiers de taille précise (utiliser la taille en octets pour les réservations de mémoire et les alignements).

types d'entier relatif		taille		types d'entier naturel	
synonyme sur Arm 32 bits	type	bits	octets : sizeof(type)	type	synonyme sur Arm 32 bits
char	int8_t	8	1	uint8_t	unsigned char
short	int16_t	16	2	uint16_t	unsigned short
int	int32_t	32	4	uint32_t	unsigned int

Annexe C : Fonction de multiplication

Voici une réalisation possible raisonnablement efficace de la fonction de multiplication :

```
1  @ Algorithme de multiplication par addition et decalage
2  @ Principe : pour chaque bit i de a  valant 1, ajouter   b << i
3  @ unsigned int mult (unsigned int a, unsigned int b) {
4  @   unsigned int resultat=0;
5  @   while (a != 0) {
6  @     if ((a & 1) != 0) {// ajouter b si a_0 == 1
7  @        resultat = resultat + b;}
8  @     b = b << 1; a = a >> 1;}
9  @   return resultat; }
10
11 @ Convention appel :
12 @ en entree, a : r0 ; b : r1 ; parametre temporaire, resultat : r2;
13 @ valeur retour : r0 (attention le r0 en entree est ecrase)
14    .text
15    .global   mult
16 mult:
17    stmfd sp!, {r1,r2}
18    mov   r2,#0
19    b     ctq
20 tq: tst   r0,#1          @ (a&1)
21    addne r2,r2,r1        @ ajouter b si a_0 == 1
22    mov   r1,r1, LSL #1
23    mov   r0,r0, LSR #1
24 ctq: cmp r0,#0
25    bne   tq
26    mov   r0,r2           @ return resultat
27    ldmfd sp!,{r1,r2}
28    bx lr
```

Annexe D : le processeur jouet à 5 instructions



FIGURE 12 – Partie opérative



FIGURE 13 – Partie automate de contrôle

Annexe E : Spécification des fonctions d'entrée/sortie définies dans `es.s`

```
1  @ fichier es.s
2  @ fonctions entrees sorties
3
4          .global AlaLigne
5          .global EcrCar
6          .global EcrChn
7          .global EcrChaine
8          .global EcrHexa32
9          .global EcrHexa16
10         .global EcrHexa8
11         .global EcrZdecimal32
12         .global EcrZdecimal16
13         .global EcrZdecimal8
14         .global EcrNdecim32
15         .global EcrNdecimal32
16         .global EcrNdecimal16
17         .global EcrNdecimal8
18         .global Lire32
19         .global Lire16
20         .global Lire8
21         .global LireCar
22         .global LireChaine
23
24 @ Extraits
25
26 @ AlaLigne :
27 @    retour a la ligne
28
29 @ EcrCar :
30 @    ecriture caractère dont la valeur est dans r1
31
32 @ EcrChaine :
33 @    ecriture de la chaine avec adresse  dans r1
34
35 @ EcrHexa32 :
36 @    ecriture mot de 32 bits en hexadécimal
37 @    la valeur a afficher est dans r1
38
39 @ EcrZdecimal32 :
40 @    ecriture en decimal entier relatif represente sur 32 bits
41 @    entier  dans r1
42
43 @ EcrZdecimal8 :
```

```
@    ecriture en decimal entier relatif represente sur 8 bits
@    entier dans les 8 bits de poids faibles de r1
@    attention : les bits 15 a 8 de r1 sont eventuellement modifies

@ EcrNdecim32 :
@    ecriture sans retour à la ligne entier naturel represente sur 32 bits
@    entier dans r1

@ EcrNdecimal32 :
@    ecriture en decimal entier naturel represente sur 32 bits
@    entier dans r1

@ EcrNdecimal8 :
@    ecriture en decimal entier naturel represente sur 8 bits
@    entier dans les 8 bits de poids faibles de r1
@    attention : les bits 15 a 8 de r1 sont mis a 0

@ Lire32 :
@    lecture entier represente sur 32 bits
@    adresse entier doit etre donnee dans r1

@ Lire8 :
@    lecture entier represente sur 8 bits
@    adresse entier doit etre donnee dans r1

@ LireCar :
@    lecture caractere tape au clavier
@    adresse caractere (code en ascii) doit etre donnee dans r1

@ LireChaine :
@    lecture  chaine de caractere tapee au clavier (max 80 car.)
@    adresse de la chaine donnee dans r1
```

Annexe F : Memento Arm

Organisation des registres

Dans le mode dit “utilisateur” le processeur `Arm` a 16 registres visibles de taille 32 bits :

— `arm-eabi-gcc` utilise `r0, r1, r2 et r3` pour le les paramètres et résultat de fonction.
— les registres `r4, r5, ..., r10` doivent pouvoir être utilisés librement par le programmeur.
— `arm-eabi-gcc` utilise `r11` (`fp` ou ”frame pointer”) pour l’environnement des fonctions.
— le registre `r12` (synonyme `ip`) est un “intra-procedure call scratch register”.
— `r13` (synonyme `sp`, comme “stack pointer”) est utilisé comme registre pointeur de pile.
— `r14` (synonyme `lr` comme “link register”) est utilisé par `bl` pour l’adresse de retour.
— `r15` (synonyme `pc`, comme “program counter”) est le registre compteur de programme.

Le processeur a de plus un registre d’état qui comporte les codes de conditions arithmétiques `N, Z, C` et `V`.

Instructions du processeur Arm

Nom	Explication du nom	Opération	Remarque
`AND`	AND	et bit à bit	
`EOR`	Exclusive OR	ou exclusif bit à bit	
`SUB`	SUBstract	soustraction	
`RSB`	Reverse SuBstract	soustraction inversée	
`ADD`	ADDition	addition	
`ADC`	ADdition with Carry	addition avec retenue	
`SBC`	SuBstract with Carry	soustraction avec emprunt	
`RSC`	Reverse Substract with Carry	soustraction inversée avec emprunt	
`TST`	TeST	et bit à bit	pas `rd`
`TEQ`	Test EQuivalence	ou exclusif bit à bit	pas `rd`
`CMP`	CoMPare	soustraction	pas `rd`
`CMN`	CoMpare Not	addition	pas `rd`
`ORR`	OR	ou bit à bit	
`MOV`	MOVe	copie	pas `rn`
`BIC`	BIt Clear	et not bit à bit	
`MVN`	MoVe Not	not (complément à 1)	pas `rn`
`MUL`	MULtiplication	multiplication tronquée	pas `r15`
`B**`	Branch	rupture de séquence conditionnelle	** = condition, cf. ci-dessous
`BL`	Branch and Link	appel sous-programme	adresse de retour dans r14
`LDR`	Load Register	lecture mémoire	
`STR`	Store Register'	écriture mémoire	

L'opérande source d'une instruction `MOV` peut être une valeur immédiate notée `#5` ou un registre noté `Ri`, `i` désignant le numéro du registre. Il peut aussi être le contenu d'un registre sur lequel on applique un décalage de `k` bits ; on note `Ri, DEC #k`, avec `DEC` $\in$ {`LSL`, `LSR`, `ASR`, `ROR`}.

Codes conditions du processeur Arm

La table suivante donne les codes de conditions arithmétiques `**` pour l'instruction de rupture de séquence `B**`.

mnémonique	signification	condition testée
EQ	égal	Z
NE	non égal	$\overline{Z}$
CS/HS	$\geq$ dans N	C
CC/LO	$<$ dans N	$\overline{C}$
MI	moins	N
PL	plus	$\overline{N}$
VS	débordement	V
VC	pas de débordement	$\overline{V}$
HI	$>$ dans N	$C \wedge \overline{Z}$
LS	$\leq$ dans N	$\overline{C} \vee Z$
GE	$\geq$ dans Z	$(N \wedge V) \vee (\overline{N} \wedge \overline{V})$
LT	$<$ dans Z	$(N \wedge \overline{V}) \vee (\overline{N} \wedge V)$
GT	$>$ dans Z	$\overline{Z} \wedge ((N \wedge V) \vee (\overline{N} \wedge \overline{V}))$
LE	$\leq$ dans Z	$Z \vee (N \wedge \overline{V}) \vee (\overline{N} \wedge V)$
AL	toujours	

Registres du processeur Arm

Registre	Spécial	Sauvegarde	Rôle dans l'appel de procédure
`r15`	`pc`	-	*Program Counter*
`r14`	`lr`	appelée	*Link Register*
`r13`	`sp`	appelée	*Stack Pointer*
`r12`	`ip`	appelée	*Intra-Procedure-call scratch register*
`r11`	`fp`	appelée	*Frame Pointer* ou variable 8
`r10`		appelée	variable 7
`r9`		appelée	variable 6
`r8`		appelée	variable 5
`r7`		appelée	variable 4
`r6`		appelée	variable 3
`r5`		appelée	variable 2
`r4`		appelée	variable 1
`r3`		appelante	argument 4
`r2`		appelante	argument 3
`r1`		appelante	argument 2 / résultat
`r0`		appelante	argument 1 / résultat

Convention d'appel Arm

— les 4 premiers paramètres sont passés dans les registres `R0` à `R3`
  s'il y a plus de 4 paramètres, les suivants sont stockés dans la pile
— la valeur de retour de la fonction est stockée dans `R0` (et `R1` si besoin)
— certains registres à sauvegarder par l'**appelante**, d'autres par l'**appelée**